﻿/**
 * @author sminutoli
 * versión 1.0
 * revisar el chequeo cabezon que hay ahí abajo
 * 
 */
import com.dinmax.flash.draw.cabeza.ButtonCabezaTooltip
import com.dinmax.flash.central.bridge.DATABridge
import com.dinmax.flash.central.bridge.EXTENSIONBridge
import com.dinmax.flash.draw.cabeza.ControlCabeza

class com.dinmax.flash.central.bridge.EXTENSIONButtonsBridge
extends MovieClip
implements com.dinmax.flash.draw.cabeza.SelectedListener {
	
	private var mcUserButton:ButtonCabezaTooltip;
	private var mcLogButton:ButtonCabezaTooltip;
	private var mcMessagesButton:ButtonCabezaTooltip;
	private var mcPreferencesButton:ButtonCabezaTooltip;
	private var mcVoiceMailsButton:ButtonCabezaTooltip;
	private var mcLightMessages:ButtonCabezaTooltip; // luz de mensajitos de texto
	private var mcLightVoiceMails:ButtonCabezaTooltip; // luz de mensajitos de voz
	
	function onLoad():Void {
		init();
	};
	
	function init():Void {
		//
		var buttons:Array = new Array();
		var tooltips:Array = new Array();
		buttons=[mcUserButton, mcLogButton, mcMessagesButton, mcVoiceMailsButton, mcPreferencesButton];
		tooltips=["info usuario", "ver logs", "mensajes de texto", "mensajes de voz", "preferencias"];
		//

		// itero en los arrays y llamo a la funci
		for(var i:Number=0; i < buttons.length; i++){
			buttonEvent(buttons[i], tooltips[i]);
		};
		
		mcPreferencesButton.setDisabled();
		
		//este chequeo es bastante cabezon
		if(! (this._parent.getBridge().getExtension(this._parent).getTextMessages().length > 0) && ! this._parent.isExtensionAutorized()) mcMessagesButton.setDisabled();
		if(! (this._parent.getBridge().getExtension(this._parent).getVoiceMessages().length > 0)) mcVoiceMailsButton.setDisabled();
		//
		
		this._parent.registerChild(this);
	};
	
	private function buttonEvent(button:ButtonCabezaTooltip, tooltip:String):Void {
		button.addListener(this);
		button.setToolTipMsg(tooltip);
		button.setToolTipTimer(150);
	};
	
	public function lightMessage(lightOn:Boolean):Void {
		//trace("EXTENSIONButtonsBridge.as | lightMessage()");
		if(lightOn){
			this.mcLightMessages.gotoAndPlay("lightOn");
		} else {
			this.mcLightMessages.gotoAndStop(1);
		}
	};
	
	public function lightVoiceMail(lightOn:Boolean):Void {
		if(lightOn){
			this.mcLightVoiceMails.gotoAndPlay("lightOn");
		} else {
			this.mcLightVoiceMails.gotoAndStop(1);
		}
	};
	
	function selected(control:ControlCabeza):Void {
		switch(control){
			case this.mcUserButton:
				this._parent.EXTENSIONBridge_2_CENTRALBridge_getUser();
				this._parent.createData(DATABridge.USER);
			break;
			
			case this.mcLogButton:
				this._parent.EXTENSIONBridge_2_CENTRALBridge_getLog();
				this._parent.createData(DATABridge.LOG);
			break;
			
			case this.mcMessagesButton:
				//this._parent.EXTENSIONBridge_2_CENTRALBridge_getMessages();
				this._parent.createData(DATABridge.MESSAGES);
			break;
			
			case this.mcVoiceMailsButton:
				//this._parent.EXTENSIONBridge_2_CENTRALBridge_getMessages();
				this._parent.createData(DATABridge.VOICEMAILS);
			break;
			
			case this.mcPreferencesButton:
			break;
			
		};
	};
}